
    BDM

Elixir module that implements the Block Decomposition Method (BDM) developed by Hector Zenil et al to approximate the algorithmic complexity of datasets by decomposing them into smaller blocks and using precomputed CTM (Coding Theorem Method) values.
This implementation is based on PyBDM by Szymon Talaga et al.
Installation
The package can be installed by adding bdm to your list of dependencies in mix.exs:
def deps do
  [
    {:bdm, "~> 0.5.0"}
  ]
end
Key Features
	Supports both 1D (binary strings) and 2D (binary matrices) data
	Two boundary conditions: :ignore and :correlated	Ignore: Discards incomplete blocks
	Correlated: Uses sliding window approach with fixed window size of 1


	Backend could be selected for BDM using the backend option. The default is CTM (:ctm), Lempel–Ziv Complexity (:lzc) [Kaspar, Schuster, 1987] is a performant, but less accurate alternative. Useful for large data sets
	Precomputed CTM values for small binary strings and matrices	1D lists, with maximum block size 11
	2D matrices with maximum block size 4*4


	Fallback mechanism for missing values (max CTM + 1 bit)

Main Modules and Functions
	BDM.new/6 - Creates a new BDM analysis structure
	BDM.compute/2 - Computes the BDM complexity of a dataset. The input argument can either be a 1D list, a 2D matrix (list of lists) or an %Nx.Tensor{}
	BDM.PerturbationAnalysis - Performs perturbation analysis, identifies complexity-driving elements	single_bit_perturbations/1 - Single-bit flip perturbations
	random_perturbations/3- Random noise perturbations with configurable noise levels
	calculate_perturbation_effects/3, perturbation_landscape/3 - More insight into the perturbation behavior
	sensitivity_profile/2, detect_critical_positions/2, stability_coefficient/4 - Comprehensive sensitivity analysis


	BDM.MinimalInformationLoss - Minimal Information Loss algorithm for feature selection and dimension reduction	feature_scores/2, select_features/3 - Feature scoring based on the change in the object’s estimated algorithmic information content
	reduce_dimensions/2 - Dimension reduction based on feature selection



Usage
bdm = BDM.new(2, 2, 2)

large_matrix = [
  [0, 1, 0, 1, 0, 1],
  [1, 0, 1, 0, 1, 0],
  [0, 1, 0, 1, 0, 1],
  [1, 0, 1, 0, 1, 0],
  [0, 1, 0, 1, 0, 1],
  [1, 0, 1, 0, 1, 0]
]

complexity_2x2 = BDM.compute(bdm, large_matrix)
For more details and explanations, check the Livebook.
How BDM Works
The Block Decomposition Method decomposes an object into smaller parts for which there exist, thanks to CTM, good approximations to their algorithmic complexity, and then aggregates these quantities by following the rules of algorithmic information theory.
Foundation: Coding Theorem Method (CTM)
The Coding Theorem Method (CTM) is a numerical approximation to the algorithmic complexity of single objects.
BDM builds upon the Coding Theorem Method (CTM), which approximates algorithmic complexity using this formula:
$$
K(s) ≈ -log_2(P(s))
$$
where P(s) is the algorithmic probability of string s. CTM approximates algorithmic probability by exploring spaces of Turing machines with n symbols and m states, counting how many produce a given output, and dividing by the total number of machines that halt.
The BDM Process
The Block Decomposition Method operates in three main stages: decomposition, lookup, and aggregation.
Step 1: Precomputation
First precompute CTM values for all possible small objects of a given type (e.g. all binary strings of up to 12 digits or all possible square binary matrices up to 4x4) and store them in an efficient lookup table.
Step 2: Decomposition
Any arbitrarily large object can be decomposed into smaller slices of appropriate sizes for which CTM values can be looked up very fast. The method partitions the input data into blocks of predetermined sizes.
Step 3: Lookup
For each unique slice created during decomposition, the method looks up the precomputed CTM value from the lookup table.
Step 4: Aggregation
The CTM values for slices can be aggregated back to a global estimate of Kolmogorov complexity for the entire object using the BDM formula:
$$
BDM(X) = \sum_i{CTM(sᵢ) + log_2(nᵢ)}
$$
where:
	$i$ indexes the set of all unique slices
	$CTM(sᵢ)$ is the complexity of slice $i$
	$nᵢ$ is the number of occurrences of slice $i$

Boundary Conditions
If the object’s size is not a multiple of the block size, the boundary (or residual) region remains. This module handles two boundary conditions:
	Ignore: Discard incomplete blocks at the edges
	Correlated: Use sliding window instead of slicing. By choosing the right window size, no residual blocks will remain at the boundary

Key Advantages
	Computational Efficiency: Instead of computing CTM for each large dataset (which is extremely expensive), BDM uses precomputed values for small blocks
	Scalability: Can handle arbitrarily large datasets by decomposing them into manageable pieces
	Practical Approximation: Provides a computable approximation to the theoretically uncomputable Kolmogorov complexity

The method essentially transforms an intractable global computation into a series of fast local lookups, making algorithmic complexity estimation practical for real-world datasets.
Citations
	Soler-Toscano F., Zenil H., Delahaye J.-P. and Gauvrit N. (2014) Calculating Kolmogorov Complexity from the Output Frequency Distributions of Small Turing Machines. PLoS ONE 9(5): e96223.
	Zenil H., Soler-Toscano F., Kiani N.A., Hernández-Orozco S., Rueda-Toicen A. (2016) A Decomposition Method for Global Evaluation of Shannon Entropy and Local Estimations of Algorithmic Complexity. arXiv:1609.00110
	Zenil H, Kiani NA, Tegnér J. Algorithmic Information Dynamics: A Computational Approach to Causality with Applications to Living Systems. Cambridge University Press; 2023.
	F. Kaspar, H. G. Schuster (1987) Easily calculable measure for the complexity of spatiotemporal patterns. Phys. Rev. A 36, 842
	Hector Zenil and Narsis A. Kiani and Alyssa Adams and Felipe S. Abrahão and Antonio Rueda-Toicen and Allan A. Zea and Luan Ozelim and Jesper Tegnér (2018) Minimal Algorithmic Information Loss Methods for Dimension Reduction, Feature Selection and Network Sparsification. arXiv:1802.05843



  

    LICENSE


                                 Apache License
                           Version 2.0, January 2004
                        http://www.apache.org/licenses/

   TERMS AND CONDITIONS FOR USE, REPRODUCTION, AND DISTRIBUTION

   1. Definitions.

      "License" shall mean the terms and conditions for use, reproduction,
      and distribution as defined by Sections 1 through 9 of this document.

      "Licensor" shall mean the copyright owner or entity authorized by
      the copyright owner that is granting the License.

      "Legal Entity" shall mean the union of the acting entity and all
      other entities that control, are controlled by, or are under common
      control with that entity. For the purposes of this definition,
      "control" means (i) the power, direct or indirect, to cause the
      direction or management of such entity, whether by contract or
      otherwise, or (ii) ownership of fifty percent (50%) or more of the
      outstanding shares, or (iii) beneficial ownership of such entity.

      "You" (or "Your") shall mean an individual or Legal Entity
      exercising permissions granted by this License.

      "Source" form shall mean the preferred form for making modifications,
      including but not limited to software source code, documentation
      source, and configuration files.

      "Object" form shall mean any form resulting from mechanical
      transformation or translation of a Source form, including but
      not limited to compiled object code, generated documentation,
      and conversions to other media types.

      "Work" shall mean the work of authorship, whether in Source or
      Object form, made available under the License, as indicated by a
      copyright notice that is included in or attached to the work
      (an example is provided in the Appendix below).

      "Derivative Works" shall mean any work, whether in Source or Object
      form, that is based on (or derived from) the Work and for which the
      editorial revisions, annotations, elaborations, or other modifications
      represent, as a whole, an original work of authorship. For the purposes
      of this License, Derivative Works shall not include works that remain
      separable from, or merely link (or bind by name) to the interfaces of,
      the Work and Derivative Works thereof.

      "Contribution" shall mean any work of authorship, including
      the original version of the Work and any modifications or additions
      to that Work or Derivative Works thereof, that is intentionally
      submitted to Licensor for inclusion in the Work by the copyright owner
      or by an individual or Legal Entity authorized to submit on behalf of
      the copyright owner. For the purposes of this definition, "submitted"
      means any form of electronic, verbal, or written communication sent
      to the Licensor or its representatives, including but not limited to
      communication on electronic mailing lists, source code control systems,
      and issue tracking systems that are managed by, or on behalf of, the
      Licensor for the purpose of discussing and improving the Work, but
      excluding communication that is conspicuously marked or otherwise
      designated in writing by the copyright owner as "Not a Contribution."

      "Contributor" shall mean Licensor and any individual or Legal Entity
      on behalf of whom a Contribution has been received by Licensor and
      subsequently incorporated within the Work.

   2. Grant of Copyright License. Subject to the terms and conditions of
      this License, each Contributor hereby grants to You a perpetual,
      worldwide, non-exclusive, no-charge, royalty-free, irrevocable
      copyright license to reproduce, prepare Derivative Works of,
      publicly display, publicly perform, sublicense, and distribute the
      Work and such Derivative Works in Source or Object form.

   3. Grant of Patent License. Subject to the terms and conditions of
      this License, each Contributor hereby grants to You a perpetual,
      worldwide, non-exclusive, no-charge, royalty-free, irrevocable
      (except as stated in this section) patent license to make, have made,
      use, offer to sell, sell, import, and otherwise transfer the Work,
      where such license applies only to those patent claims licensable
      by such Contributor that are necessarily infringed by their
      Contribution(s) alone or by combination of their Contribution(s)
      with the Work to which such Contribution(s) was submitted. If You
      institute patent litigation against any entity (including a
      cross-claim or counterclaim in a lawsuit) alleging that the Work
      or a Contribution incorporated within the Work constitutes direct
      or contributory patent infringement, then any patent licenses
      granted to You under this License for that Work shall terminate
      as of the date such litigation is filed.

   4. Redistribution. You may reproduce and distribute copies of the
      Work or Derivative Works thereof in any medium, with or without
      modifications, and in Source or Object form, provided that You
      meet the following conditions:

      (a) You must give any other recipients of the Work or
          Derivative Works a copy of this License; and

      (b) You must cause any modified files to carry prominent notices
          stating that You changed the files; and

      (c) You must retain, in the Source form of any Derivative Works
          that You distribute, all copyright, patent, trademark, and
          attribution notices from the Source form of the Work,
          excluding those notices that do not pertain to any part of
          the Derivative Works; and

      (d) If the Work includes a "NOTICE" text file as part of its
          distribution, then any Derivative Works that You distribute must
          include a readable copy of the attribution notices contained
          within such NOTICE file, excluding those notices that do not
          pertain to any part of the Derivative Works, in at least one
          of the following places: within a NOTICE text file distributed
          as part of the Derivative Works; within the Source form or
          documentation, if provided along with the Derivative Works; or,
          within a display generated by the Derivative Works, if and
          wherever such third-party notices normally appear. The contents
          of the NOTICE file are for informational purposes only and
          do not modify the License. You may add Your own attribution
          notices within Derivative Works that You distribute, alongside
          or as an addendum to the NOTICE text from the Work, provided
          that such additional attribution notices cannot be construed
          as modifying the License.

      You may add Your own copyright statement to Your modifications and
      may provide additional or different license terms and conditions
      for use, reproduction, or distribution of Your modifications, or
      for any such Derivative Works as a whole, provided Your use,
      reproduction, and distribution of the Work otherwise complies with
      the conditions stated in this License.

   5. Submission of Contributions. Unless You explicitly state otherwise,
      any Contribution intentionally submitted for inclusion in the Work
      by You to the Licensor shall be under the terms and conditions of
      this License, without any additional terms or conditions.
      Notwithstanding the above, nothing herein shall supersede or modify
      the terms of any separate license agreement you may have executed
      with Licensor regarding such Contributions.

   6. Trademarks. This License does not grant permission to use the trade
      names, trademarks, service marks, or product names of the Licensor,
      except as required for reasonable and customary use in describing the
      origin of the Work and reproducing the content of the NOTICE file.

   7. Disclaimer of Warranty. Unless required by applicable law or
      agreed to in writing, Licensor provides the Work (and each
      Contributor provides its Contributions) on an "AS IS" BASIS,
      WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or
      implied, including, without limitation, any warranties or conditions
      of TITLE, NON-INFRINGEMENT, MERCHANTABILITY, or FITNESS FOR A
      PARTICULAR PURPOSE. You are solely responsible for determining the
      appropriateness of using or redistributing the Work and assume any
      risks associated with Your exercise of permissions under this License.

   8. Limitation of Liability. In no event and under no legal theory,
      whether in tort (including negligence), contract, or otherwise,
      unless required by applicable law (such as deliberate and grossly
      negligent acts) or agreed to in writing, shall any Contributor be
      liable to You for damages, including any direct, indirect, special,
      incidental, or consequential damages of any character arising as a
      result of this License or out of the use or inability to use the
      Work (including but not limited to damages for loss of goodwill,
      work stoppage, computer failure or malfunction, or any and all
      other commercial damages or losses), even if such Contributor
      has been advised of the possibility of such damages.

   9. Accepting Warranty or Additional Liability. While redistributing
      the Work or Derivative Works thereof, You may choose to offer,
      and charge a fee for, acceptance of support, warranty, indemnity,
      or other liability obligations and/or rights consistent with this
      License. However, in accepting such obligations, You may act only
      on Your own behalf and on Your sole responsibility, not on behalf
      of any other Contributor, and only if You agree to indemnify,
      defend, and hold each Contributor harmless for any liability
      incurred by, or claims asserted against, such Contributor by reason
      of your accepting any such warranty or additional liability.

   END OF TERMS AND CONDITIONS

   APPENDIX: How to apply the Apache License to your work.

      To apply the Apache License to your work, attach the following
      boilerplate notice, with the fields enclosed by brackets "[]"
      replaced with your own identifying information. (Don't include
      the brackets!)  The text should be enclosed in the appropriate
      comment syntax for the file format. We also recommend that a
      file or class name and description of purpose be included on the
      same "printed page" as the copyright notice for easier
      identification within third-party archives.

   Copyright [yyyy] [name of copyright owner]

   Licensed under the Apache License, Version 2.0 (the "License");
   you may not use this file except in compliance with the License.
   You may obtain a copy of the License at

       http://www.apache.org/licenses/LICENSE-2.0

   Unless required by applicable law or agreed to in writing, software
   distributed under the License is distributed on an "AS IS" BASIS,
   WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
   See the License for the specific language governing permissions and
   limitations under the License.



  

    Block Decomposition Method (BDM)

Mix.install([
  {:nx, "~> 0.6"},
  {:kino, "~> 0.11"},
  {:explorer, "~> 0.7"},
  {:vega_lite, "~> 0.1"},
  {:kino_vega_lite, "~> 0.1"},
  {:bdm, "~> 0.3.1"}
])

alias VegaLite, as: Vl
Introduction
The Block Decomposition Method (BDM) is a computational approach developed by Hector Zenil and colleagues to approximate the algorithmic complexity of datasets. This method extends the Coding Theorem Method (CTM) to handle larger datasets by decomposing them into smaller blocks and aggregating their complexities.
Key Concepts
	Algorithmic complexity (also known as Kolmogorov complexity): Measures the length of the shortest computer program that can generate a given dataset. While this is theoretically uncomputable, BDM provides a practical approximation.
	Algorithmic Probability: Theoretical foundation based on Solomonoff-Levin theory. The algorithmic probability of an object x is the probability AP of a binary computer program p producing x by chance running on a Turing-complete computer language L and halting.
	Coding Theorem Method (CTM): Uses precomputed output frequency distributions from Turing machines
	Block Decomposition: Breaks large datasets into manageable blocks

Interactive Examples
Example 1: Simple Binary Sequences
random_seq = for _ <- 1..200, do: Enum.random([0, 1])
ordered_seq = List.duplicate([0, 1], 100) |> List.flatten()
constant_seq = List.duplicate(0, 200)
pattern_seq = [0, 0, 1] |> List.duplicate(70) |> List.flatten() |> Enum.take(200)

bdm = BDM.new(1, 2, 3, :ignore)

Kino.DataTable.new(
  [
    %{"Pattern" => "random", "BDM" => BDM.compute(bdm, random_seq)},
    %{"Pattern" => "ordered", "BDM" => BDM.compute(bdm, ordered_seq)},
    %{"Pattern" => "constant", "BDM" => BDM.compute(bdm, constant_seq)},
    %{"Pattern" => "pattern", "BDM" => BDM.compute(bdm, pattern_seq)}
  ],
  keys: ["Pattern", "BDM"],
  name: "BDM of binary sequences"
)
Example 2: 2D Pattern Analysis
defmodule TestData do
  def create_2d_pattern(:random, size) do
      for _ <- 1..size, do: for(_ <- 1..size, do: Enum.random([0, 1]))
  end

  def create_2d_pattern(:checkerboard, size) do
    for i <- 0..(size-1) do
      for j <- 0..(size-1) do
        rem(i + j, 2)
      end
    end
  end

  def create_2d_pattern(:diagonal, size) do
    for i <- 0..(size-1) do
      for j <- 0..(size-1) do
        if i == j, do: 1, else: 0
      end
    end
  end
 
  def create_2d_pattern(:constant, size) do
    for _ <- 1..size, do: List.duplicate(0, size)
  end
end

patterns = %{
  random: TestData.create_2d_pattern(:random, 8),
  checkerboard: TestData.create_2d_pattern(:checkerboard, 8),
  diagonal: TestData.create_2d_pattern(:diagonal, 8),
  constant: TestData.create_2d_pattern(:constant, 8)
}
defmodule Visualizer do
  def create_heatmap(image, title) do
    data = image
    |> Enum.with_index()
    |> Enum.flat_map(fn {row, i} ->
      row
      |> Enum.with_index()
      |> Enum.map(fn {value, j} ->
        %{x: j, y: i, value: value, image: title}
      end)
    end)

    Vl.new(width: 200, height: 200, title: title)
    |> Vl.data_from_values(data)
    |> Vl.mark(:rect)
    |> Vl.encode_field(:x, "x", type: :ordinal, title: "Column")
    |> Vl.encode_field(:y, "y", type: :ordinal, title: "Row", sort: :descending)
    |> Vl.encode_field(:color, "value", type: :quantitative, scale: [scheme: "viridis"], title: "Value")
  end
end

random = Visualizer.create_heatmap(patterns.random, "Random")
checkerboard = Visualizer.create_heatmap(patterns.checkerboard, "Checkerboard")
diagonal = Visualizer.create_heatmap(patterns.diagonal, "Diagonal")
constant = Visualizer.create_heatmap(patterns.constant, "Constant")

Kino.Layout.grid([random, checkerboard, diagonal, constant], columns: 2)
Calculate BDM for each pattern:
bdm = BDM.new(2, 2, 2, :ignore)

pattern_viz = 
  patterns
  |> Enum.map(fn {name, pattern} ->
    bdm_value = BDM.compute(bdm, pattern)
    %{"Pattern" => Atom.to_string(name), "BDM" => bdm_value}
  end)

Kino.DataTable.new(
  pattern_viz,
  keys: ["Pattern", "BDM"],
  name: "BDM of 2D patterns"
)
Example 3: Interactive Visualization
bdm = BDM.new(1, 2, 3, :ignore)

sequence_lengths = [10, 20, 30, 40, 50]
complexity_data = 
  for len <- sequence_lengths do
    random_seq = for _ <- 1..len, do: Enum.random([0, 1])
    ordered_seq =
      List.duplicate([0, 1], div(len, 2))
      |> List.flatten()
      |> Enum.take(len)
    constant_seq = List.duplicate(0, len)
    
    %{
      length: len,
      random: BDM.compute(bdm, random_seq),
      ordered: BDM.compute(bdm, ordered_seq),
      constant: BDM.compute(bdm, constant_seq)
    }
  end

viz_data = 
  complexity_data
  |> Enum.flat_map(fn row ->
    [
      %{length: row.length, complexity: row.random, type: "Random"},
      %{length: row.length, complexity: row.ordered, type: "Ordered"},
      %{length: row.length, complexity: row.constant, type: "Constant"}
    ]
  end)

Vl.new(width: 600, height: 400)
|> Vl.data_from_values(viz_data)
|> Vl.mark(:line, point: true)
|> Vl.encode_field(:x, "length", type: :quantitative, title: "Sequence Length")
|> Vl.encode_field(:y, "complexity", type: :quantitative, title: "BDM Complexity")
|> Vl.encode_field(:color, "type", type: :nominal, title: "Sequence Type")
|> Vl.config(legend: [orient: "top"])
Advanced Features
Block Size Analysis
Analyzing how block size affects BDM estimation:
block_sizes = 2..12
block_analysis_viz = 
  for block_size <- block_sizes do
    bdm = BDM.new(1, 2, block_size, :ignore)
    bdm_value = BDM.compute(bdm, pattern_seq)
    %{"Block Size" => block_size, "BDM" => bdm_value}
  end

Kino.DataTable.new(
  block_analysis_viz,
  keys: ["Block Size", "BDM"],
  name: "Block Size Analysis"
)
Vl.new(width: 800, height: 400, title: "Block Size Analysis")
|> Vl.data_from_values(block_analysis_viz)
|> Vl.mark(:line, point: true)
|> Vl.encode_field(:x, "Block Size", type: :quantitative)
|> Vl.encode_field(:y, "BDM", type: :quantitative)
|> Vl.config(legend: [orient: "top"])
Comparative Analysis with Shannon Entropy and LZW
Utility modules:
defmodule Entropy do
  def shannon_entropy(data) do
    frequencies = Enum.frequencies(data)
    total = length(data)
    
    frequencies
    |> Enum.map(fn {_, count} ->
      p = count / total
      -p * :math.log2(p)
    end)
    |> Enum.sum()
  end
end

defmodule LZW do
  def compress_list(data) when is_list(data) do
    unique_elements = data |> Enum.uniq()
    initial_dict = 
      unique_elements
      |> Enum.with_index()
      |> Enum.into(%{}, fn {elem, idx} -> {[elem], idx} end)
    
    next_code = map_size(initial_dict)
    
    {result, _dict, _code} = compress_helper(data, [], initial_dict, next_code, [])
    Enum.reverse(result)
  end

  def compress_matrix(matrix) when is_list(matrix) do
    rows = length(matrix)
    cols = if rows > 0, do: length(hd(matrix)), else: 0
    
    flat_data = List.flatten(matrix)
    compressed = compress_list(flat_data)
    
    {compressed, {rows, cols}}
  end

  defp compress_helper([], current_string, dict, next_code, result) do
    case current_string do
      [] -> {result, dict, next_code}
      _ -> 
        code = Map.get(dict, current_string)
        {[code | result], dict, next_code}
    end
  end

  defp compress_helper([char | rest], current_string, dict, next_code, result) do
    new_string = current_string ++ [char]
    
    case Map.get(dict, new_string) do
      nil ->
        current_code = Map.get(dict, current_string)
        new_dict = Map.put(dict, new_string, next_code)
        compress_helper(rest, [char], new_dict, next_code + 1, [current_code | result])
      
      _code ->
        compress_helper(rest, new_string, dict, next_code, result)
    end
  end
end
Comparison:
bdm = BDM.new(1, 2, 3, :ignore)

comparison_data = [
  {constant_seq, "constant"},
  {random_seq, "random"},
  {ordered_seq, "alternating"},
  {pattern_seq, "pattern 001"}
]

data_viz =
  for {sequence, name} <- comparison_data do
    bdm_val = BDM.compute(bdm, sequence)
    entropy_val = Entropy.shannon_entropy(sequence)
    lzw_val = length(LZW.compress_list(sequence)) / length(sequence)
    %{
      "Pattern" => name,
      "BDM" => Float.round(bdm_val, 2),
      "Entropy" => Float.round(entropy_val, 2),
      "LZW" => Float.round(lzw_val, 2),
    }
  end

Kino.DataTable.new(
  data_viz,
  keys: ["Pattern", "BDM", "Entropy", "LZW"],
  name: "BDM vs Shannon Entropy and LZW Comparison"
)
Real-World Applications
Cellular Automata Analysis
defmodule CellularAutomata do
  def rule_30(initial_state, steps) do
    evolve(initial_state, steps, &rule_30_step/1)
  end
  
  defp rule_30_step(state) do
    len = length(state)
    for i <- 0..(len-1) do
      left = Enum.at(state, rem(i - 1 + len, len))
      center = Enum.at(state, i)
      right = Enum.at(state, rem(i + 1, len))
      
      case {left, center, right} do
        {1, 1, 1} -> 0
        {1, 1, 0} -> 0
        {1, 0, 1} -> 0
        {1, 0, 0} -> 1
        {0, 1, 1} -> 1
        {0, 1, 0} -> 1
        {0, 0, 1} -> 1
        {0, 0, 0} -> 0
      end
    end
  end
  
  defp evolve(state, 0, _rule), do: [state]
  defp evolve(state, steps, rule) do
    next_state = rule.(state)
    [state | evolve(next_state, steps - 1, rule)]
  end
end

initial_state = [0, 0, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0]
evolution = CellularAutomata.rule_30(initial_state, 20)

bdm = BDM.new(1, 2, 2, :ignore)

# Calculate BDM for each step
evolution_complexity = 
  evolution
  |> Enum.with_index()
  |> Enum.map(fn {state, step} ->
    bdm_val = BDM.compute(bdm, state)
    %{step: step, complexity: bdm_val}
  end)

viz_data = 
  evolution_complexity
  |> Enum.map(fn row ->
      %{step: row.step, complexity: Float.round(row.complexity, 2)}
  end)

Vl.new(width: 600, height: 400, title: "Cellular Automata Evolution Complexity")
|> Vl.data_from_values(viz_data)
|> Vl.mark(:line, point: true)
|> Vl.encode_field(:x, "step", type: :quantitative, title: "Step")
|> Vl.encode_field(:y, "complexity", type: :quantitative, title: "BDM Complexity")
|> Vl.config(legend: [orient: "top"])
Performance Considerations
Memory and Time Complexity
defmodule Benchmark do
  def time(func) do
    start_time = System.monotonic_time(:millisecond)
    result = func.()
    end_time = System.monotonic_time(:millisecond)
    {result, end_time - start_time}
  end
end

bdm = BDM.new(1, 2, 2, :ignore)

viz_data = for len <- 10000..100000//5000 do
  test_seq = for _ <- 1..len, do: Enum.random([0, 1])
  {_, time_ms} = Benchmark.time(fn -> BDM.compute(bdm, test_seq) end)
  %{
    "l" => len,
    "t" => time_ms
  }
end

Vl.new(width: 600, height: 400, title: "Performance Analysis")
|> Vl.data_from_values(viz_data)
|> Vl.mark(:line, point: true)
|> Vl.encode_field(:x, "l", type: :quantitative, title: "Dataset Length")
|> Vl.encode_field(:y, "t", type: :quantitative, title: "Time (ms)")
|> Vl.config(legend: [orient: "top"])
Perturbation Analysis
alias BDM.PerturbationAnalysis
Understanding Sensitivity to Changes
Perturbation analysis reveals how robust BDM complexity estimates are to small changes in the data. This is crucial for understanding the stability of complexity measures and identifying critical structures.
Interactive Perturbation Visualization
Creating 1D test patterns for perturbation analysis and analyzing sensitivity for each pattern:
test_patterns_1d = %{
  structured: [0, 1, 0, 1, 0, 1, 0, 1, 0, 1, 0, 1, 0, 1, 0, 1],
  random: Enum.take(Stream.repeatedly(fn -> Enum.random([0, 1]) end), 16),
  mostly_zeros: [0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 0],
  clustered: [1, 1, 1, 1, 0, 0, 0, 0, 1, 1, 1, 1, 0, 0, 0, 0]
}

bdm_1d = BDM.new(1, 2, 3, :ignore)

sensitivity_analyses = 
  for {name, pattern} <- test_patterns_1d do
    sensitivity_profile = PerturbationAnalysis.sensitivity_profile(bdm_1d, pattern)
    {name, pattern, sensitivity_profile}
  end

sensitivity_data =
  for {name, _, profile} <- sensitivity_analyses do
    max_sensitivity =
      profile
      |> Enum.map(&(&1.sensitivity))
      |> Enum.max()
    avg_sensitivity =
      profile
      |> Enum.map(&(&1.sensitivity))
      |> Enum.sum()
      |> Kernel./(length(profile))

    %{
      "Pattern" => Atom.to_string(name),
      "Max sensitivity" => Float.round(max_sensitivity, 3),
      "Avg sensitivity" => Float.round(avg_sensitivity, 3)
    }
  end

Kino.DataTable.new(
  sensitivity_data,
  name: "Sensitivity Analysis",
  keys: ["Pattern", "Max sensitivity", "Avg sensitivity"]
)
viz_data = 
  for {name, _pattern, profile} <- sensitivity_analyses do
    profile
    |> Enum.map(fn point ->
      Map.put(point, :pattern, to_string(name))
    end)
  end
  |> List.flatten()

Vl.new(width: 700, height: 400, title: "Position Sensitivity Analysis")
|> Vl.data_from_values(viz_data)
|> Vl.mark(:line, point: true, stroke_width: 2)
|> Vl.encode_field(:x, "position", type: :quantitative, title: "Bit Position")
|> Vl.encode_field(:y, "sensitivity", type: :quantitative, title: "Sensitivity (|ΔBDM|)")
|> Vl.encode_field(:color, "pattern", type: :nominal, title: "Pattern Type")
|> Vl.encode_field(:stroke_dash, "pattern", type: :nominal)
Noise Level Analysis
Analyze how different noise levels affect complexity:
noise_levels = [0.05, 0.1, 0.2, 0.3, 0.4, 0.5]
base_pattern = [0, 1, 0, 1, 0, 1, 0, 1] |> List.duplicate(4) |> List.flatten()

bdm = BDM.new(1, 2, 3, :ignore)

noise_analysis_data = 
  for noise_level <- noise_levels do
    # Generate multiple perturbations at this noise level
    perturbations = PerturbationAnalysis.random_perturbations(bdm, base_pattern, 20, noise_level)
    {original_bdm, results} = PerturbationAnalysis.calculate_perturbation_effects(bdm, base_pattern, perturbations)
    
    # Calculate statistics
    delta_bdms = Enum.map(results, &(&1.delta_bdm))
    mean_delta = Enum.sum(delta_bdms) / length(delta_bdms)
    std_delta = :math.sqrt(Enum.sum(Enum.map(delta_bdms, fn x -> (x - mean_delta) * (x - mean_delta) end)) / length(delta_bdms))
    
    %{
      "Noise%" => trunc(noise_level * 100),
      "BDM" => Float.round(original_bdm, 2),
      "Δ" => Float.round(std_delta, 2),
      "Mean Δ" => Float.round(mean_delta, 2),
      "Min Δ" => Float.round(Enum.min(delta_bdms), 2),
      "Max Δ" => Float.round(Enum.max(delta_bdms), 2)
    }
  end

Kino.DataTable.new(
  noise_analysis_data,
  name: "Noise Level Analysis",
  keys: ["Noise%", "BDM", "Δ", "Mean Δ", "Min Δ", "Max Δ"]
)
Noise level impact (how complexity changes scale with perturbation intensity) visualization:
viz_data = 
  noise_analysis_data
  |> Enum.flat_map(fn result ->
    noise_level = Map.get(result, "Noise%")
    [
      %{noise_level: noise_level, value: Map.get(result, "Mean Δ"), measure: "Mean Change"},
      %{noise_level: noise_level, value: Map.get(result, "Δ"), measure: "Std Deviation"},
      %{noise_level: noise_level, value: Map.get(result, "Max Δ"), measure: "Max Change"}
    ]
  end)

Vl.new(width: 600, height: 400, title: "BDM Response to Noise Levels")
|> Vl.data_from_values(viz_data)
|> Vl.mark(:line, point: true)
|> Vl.encode_field(:x, "noise_level", type: :quantitative, title: "Noise Level (%)")
|> Vl.encode_field(:y, "value", type: :quantitative, title: "BDM Change")
|> Vl.encode_field(:color, "measure", type: :nominal, title: "Measure")
|> Vl.config(legend: [orient: "top"])
Critical Point Detection
Analyze critical points (where structural changes have maximum impact) for structured pattern:
structured_pattern = [0, 1, 0, 1, 0, 1, 0, 1, 0, 1, 0, 1, 0, 1, 0, 1]
sensitivity_profile = PerturbationAnalysis.sensitivity_profile(bdm, structured_pattern)
critical_points = PerturbationAnalysis.detect_critical_positions(sensitivity_profile, 0.5)

critical_point_data =
  for point <- Enum.take(critical_points, 5) do
    %{
      "Position" => point.position,
      "Sensitivity" => Float.round(point.sensitivity, 3),
      "Relative" => Float.round(point.relative_sensitivity, 3)
    }
end

Kino.DataTable.new(
  critical_point_data,
  name: "Critical Points Analysis",
  keys: ["Position", "Sensitivity", "Relative"]
)
Perturbation Phase Space
Phase space visualization showing relationship between original complexity and perturbation effects:
viz_data = 
  for {name, pattern, _} <- sensitivity_analyses do
    original_bdm = BDM.compute(bdm, pattern)
    perturbations = PerturbationAnalysis.single_bit_perturbations(bdm, pattern)
    
    perturbations
    |> Enum.with_index()
    |> Enum.map(fn {perturbed, index} ->
      perturbed_bdm =  BDM.compute(bdm, perturbed)
      hamming_distance = Enum.zip(pattern, perturbed) 
                        |> Enum.count(fn {a, b} -> a != b end)
      
      %{
        pattern: to_string(name),
        original_bdm: original_bdm,
        perturbed_bdm: perturbed_bdm,
        delta_bdm: perturbed_bdm - original_bdm,
        hamming_distance: hamming_distance,
        position: index
      }
    end)
  end
  |> List.flatten()

Vl.new(width: 600, height: 400, title: "Perturbation Phase Space")
|> Vl.data_from_values(viz_data)
|> Vl.mark(:circle, size: 60, opacity: 0.7)
|> Vl.encode_field(:x, "original_bdm", type: :quantitative, title: "Original BDM")
|> Vl.encode_field(:y, "perturbed_bdm", type: :quantitative, title: "Perturbed BDM")
|> Vl.encode_field(:color, "pattern", type: :nominal, title: "Pattern Type")
|> Vl.encode_field(:size, "hamming_distance", type: :quantitative, title: "Hamming Distance")
Robustness Analysis
Compare robustness (which pattern types are most robust to perturbations) across different pattern types:
robustness_comparison = 
  for {name, pattern} <- test_patterns_1d do
    analysis = PerturbationAnalysis.stability_coefficient(bdm, pattern, 30, 0.15)
    Map.put(analysis, :pattern, name)
  end

viz_data = 
  robustness_comparison
  |> Enum.map(fn result ->
    %{
      pattern: to_string(result.pattern),
      stability_score: result.stability_score,
      coefficient_variation: result.coefficient_of_variation
    }
  end)

Vl.new(width: 500, height: 300, title: "Pattern Robustness Analysis")
|> Vl.data_from_values(viz_data)
|> Vl.mark(:bar)
|> Vl.encode_field(:x, "pattern", type: :nominal, title: "Pattern Type")
|> Vl.encode_field(:y, "stability_score", type: :quantitative, title: "Stability Score")
|> Vl.encode_field(:color, "pattern", type: :nominal)
Perturbation landscape
2D landscape:
bdm = BDM.new(2, 2, 2, :ignore)

landscape = PerturbationAnalysis.perturbation_landscape(bdm, patterns.random, 2)
defmodule LandscapeVisualizer do
  def extract_max_effects(landscape_result, flip_count \\ 1) do
    for point <- landscape_result do
      effect = Enum.find(point.effects, fn e -> e.num_flips == flip_count end)
      max_effect = if effect, do: effect.max_effect, else: 0.0
      max_row = if length(landscape_result) > 0 do
          landscape_result |> Enum.map(& &1.center_row) |> Enum.max()
        else
          0
        end
      size = max_row + 2  # Add 1 for 0-based indexing + 1 for radius
      
      %{
        x: point.center_col,
        y: size - 1 - point.center_row,  # Flip y coordinate
        max_effect: max_effect,
        abs_effect: abs(max_effect)
      }
    end
  end
  
  def create_landscape_heatmap(data, title) do
    Vl.new(width: 300, height: 300, title: title)
    |> Vl.data_from_values(data)
    |> Vl.mark(:rect)
    |> Vl.encode_field(:x, "x", type: :ordinal, title: "Column")
    |> Vl.encode_field(:y, "y", type: :ordinal, title: "Row")
    |> Vl.encode_field(:color, "abs_effect", 
        type: :quantitative,
        scale: [scheme: "viridis"],
        legend: [title: "Perturbation Effect"])
    |> Vl.encode(:tooltip, [
      [field: "x", type: :quantitative],
      [field: "y", type: :quantitative], 
      [field: "max_effect", type: :quantitative, title: "Max Effect"],
      [field: "abs_effect", type: :quantitative, title: "Abs Effect"]
    ])
  end
end

plot_data = LandscapeVisualizer.extract_max_effects(landscape, 1)
LandscapeVisualizer.create_landscape_heatmap(
  plot_data, 
  "Perturbation Landscape for random (1-bit flips)"
)

## [0.5.0] - 2026-01-24
### Added
- Implemented the Minimal Information Loss algorithm for feature selection and dimension reduction.

## [0.4.0] - 2025-09-13
### Added
- Backend could be selected for BDM. The default is CTM, Lempel–Ziv Complexity (LZC) is an option. Useful for large 2D matrices

## [0.3.1] - 2025-08-29
### Fixed
- Added missing CTM values

## [0.3.0] - 2025-08-29
### Changed
- Replaced the placeholder CTM values with real ones
### Removed
- Removed the incomplete implementation of the `:recursive` boundary condition
- Removed `BDM.Utils.normalize/2`

## [0.2.1] - 2025-08-27
### Added
- Added new simple 3x3 pattern type: `[[1, 0, 0], [0, 0, 0], [0, 0, 0]]` and `[0, 0, 0], [0, 0, 0], [0, 0, 1]]`

## [0.2.0] - 2025-08-26
### Added
- Using the [Nx](https://hexdocs.pm/nx) library to compute BDM for large 2D matrices (images) more efficiently

## [0.1.1] - 2025-08-21
### Fixed
- Fixed README.md 'Usage' section
- Fixed module documentation

## [0.1.0] - 2025-08-12
### Added
- Initial release

          @type t() :: %BDM{
  backend: :ctm | :lzc,
  block_size: integer(),
  boundary: boundary_condition(),
  ctm_data: map(),
  ndim: 1 | 2,
  nsymbols: integer(),
  warn_missing: boolean()
}

          @spec compute(t(), binary_string() | binary_matrix()) :: float()


      


Computes the BDM complexity of a dataset.
Parameters
	bdm: BDM instance
	data: Input data (list for 1D, Nx.Tensor or list of lists for 2D)

          @spec new(
  integer(),
  integer(),
  integer(),
  boundary_condition(),
  :ctm | :lzc,
  map() | nil,
  boolean()
) :: t()


      


Creates a new BDM instance.
Parameters
	ndim: Dimensionality (1 for strings, 2 for matrices)
	nsymbols: Number of symbols (2 for binary)
	block_size: Size of blocks for decomposition
	boundary: Boundary condition (:ignore, :correlated)
	backend: CTM (default) or LZC
	ctm_data: Optional custom CTM lookup table
	warn_missing: Whether to warn about missing CTM values

Examples
iex> BDM.new(1, 2, 2)
%BDM{ndim: 1, nsymbols: 2, block_size: 2, ctm_data: ..., warn_missing: true}

          @spec partition_1d(binary_string(), integer(), boundary_condition()) :: [
  binary_string()
]


      


Partitions 1D data into blocks according to boundary condition.

          @spec partition_2d(binary_matrix(), integer(), boundary_condition()) :: [
  binary_matrix()
]


      


Partitions 2D data into blocks according to boundary condition.

Example usage of the BDM module.

Compute Lempel–Ziv Complexity (LZC) for 2D matrices.

        Compute LZC for a binary Nx tensor or a list of lists (2D matrix).



    





      


      
        Functions

        


  
    
      
    
    
      lzc(tensor)



        
          
        

    

  


  

      

          @spec lzc(Nx.Tensor.t()) :: pos_integer()


          @spec lzc(list()) :: pos_integer()


      


Compute LZC for a binary Nx tensor or a list of lists (2D matrix).
Parameters
	tensor: a 2D Nx tensor or a lists of lists, typically binary (0/1).
Non-binary values will be stringified.

Returns
	LZC value (positive integer, the count of distinct substrings encountered)

Feature selection and dimension reduction methods based on minimizing loss of
algorithmic information, such that the estimated Kolmogorov complexity of the
reduced object remains as close as possible to that of the original.
The goal is to preserve causal/structural information, not just statistical
regularities.

        complexity(bdm, rows)

      


        Compute baseline complexity of a binary matrix.



    


    
      
        feature_scores(bdm, rows)

      


        Score each feature based on the change in the object’s estimated algorithmic
information content after the removal of the feature.



    


    
      
        reduce_dimensions(bdm, rows, k)

      


        Reduce a matrix to k features using greedy selection.



    


    
      
        select_features(bdm, rows, k)

      


        Greedy feature selection. Iteratively removing the feature with the
smallest delta until k remain.

          @spec complexity(BDM.t(), BDM.binary_matrix()) :: float()


      


Compute baseline complexity of a binary matrix.
By default we use a 2D BDM configured for binary matrices.

          @spec feature_scores(BDM.t(), BDM.binary_matrix()) :: [map()]


      


Score each feature based on the change in the object’s estimated algorithmic
information content after the removal of the feature.
Returns a list of maps:
[%{idx: 0, delta: 1.23, base: 10.0, removed: 8.77}, ...]
delta = base - removed (non-negative in most cases).

          @spec reduce_dimensions(BDM.t(), BDM.binary_matrix(), pos_integer()) :: %{
  history: [map()],
  kept: [integer()],
  reduced: list()
}


      


Reduce a matrix to k features using greedy selection.
Returns %{kept: [...], reduced: rows_k, history: [...]}.

          @spec select_features(BDM.t(), BDM.binary_matrix(), pos_integer()) ::
  {[integer()], [map()]}


      


Greedy feature selection. Iteratively removing the feature with the
smallest delta until k remain.
Returns {kept_indices, history} where history logs each removal.

Tools for analyzing how perturbations affect BDM complexity estimates in order to
identify complexity-driving elements, sensitivity, critical points and stability.

      


      
        Summary

        calculate_perturbation_effects(bdm, original_data, perturbations)

      


        Calculates BDM for original and all perturbed versions.



    


    
      
        detect_critical_positions(sensitivity_profile, threshold \\ 1.0)

      


        Identifies positions where perturbation sensitivity exceeds threshold.



    


    
      
        perturbation_landscape(bdm, data, radius)

      


        Creates a landscape showing cumulative effects of multi-bit perturbations.



    


    
      
        random_perturbations(bdm, data, num_perturbations, noise_level)

      


        Generates random perturbations with specified noise level.
noise_level: fraction of bits to flip (0.0 to 1.0)



    


    
      
        sensitivity_profile(bdm, data)

      


        Creates a sensitivity profile showing which positions are most sensitive to perturbation.



    


    
      
        single_bit_perturbations(bdm, data)

      


        Generates all possible single-bit flip perturbations of the data.



    


    
      
        stability_coefficient(bdm, data, num_trials \\ 50, noise_level \\ 0.1)

      


        Calculates stability coefficient: ratio of consistent complexity estimates.

          @spec calculate_perturbation_effects(
  BDM.t(),
  BDM.binary_string() | BDM.binary_matrix(),
  list()
) ::
  {float(), [map()]}


      


Calculates BDM for original and all perturbed versions.

          @spec detect_critical_positions([map()], number()) :: [map()]


      


Identifies positions where perturbation sensitivity exceeds threshold.

          @spec perturbation_landscape(
  BDM.t(),
  BDM.binary_string() | BDM.binary_matrix(),
  integer()
) :: [map()]


      


Creates a landscape showing cumulative effects of multi-bit perturbations.
Returns max_effect, min_effect, and avg_effect for each flip count, providing more insight
into the perturbation behavior.

          @spec random_perturbations(
  BDM.t(),
  BDM.binary_string() | BDM.binary_matrix(),
  pos_integer(),
  number()
) :: BDM.binary_string() | BDM.binary_matrix()


      


Generates random perturbations with specified noise level.
noise_level: fraction of bits to flip (0.0 to 1.0)

          @spec sensitivity_profile(BDM.t(), BDM.binary_string() | BDM.binary_matrix()) :: [
  map()
]


      


Creates a sensitivity profile showing which positions are most sensitive to perturbation.

          @spec single_bit_perturbations(BDM.t(), BDM.binary_string() | BDM.binary_matrix()) ::
  BDM.binary_string() | BDM.binary_matrix()


      


Generates all possible single-bit flip perturbations of the data.

  



    

    

  
    
      
    
    
      stability_coefficient(bdm, data, num_trials \\ 50, noise_level \\ 0.1)



        
          
        

    

  


  

      

          @spec stability_coefficient(
  BDM.t(),
  BDM.binary_string() | BDM.binary_matrix(),
  pos_integer(),
  number()
) :: map()


      


Calculates stability coefficient: ratio of consistent complexity estimates.

